Module – 3:
Arduino Scenario Based Programming

Scenario-1: Smart Street Light

- **Situation:** A street light should automatically turn ON at night and OFF during the day. If motion is detected at night, increase brightness.

- **Requirements:**
  (a) LDR → detect day/night,
  (b) PIR sensor → detect motion and
  (c) LED (PWM) → control brightness

- **Programming Task:** Write an Arduino program that:
  (a) Reads LDR value,
  (b)Turns LED ON only at night and
  (c) Increases brightness when motion is detected

Scenario-1: Smart Street Light - Solution

```
int ldrPin = A0;          // LDR connected to analog pin A0
int pirPin = 2;           // PIR sensor connected to digital pin 2
int ledPin = 9;           // LED connected to PWM pin 9

int ldrValue = 0;
int pirState = 0;

void setup() {
  pinMode(pirPin, INPUT);
  pinMode(ledPin, OUTPUT);
  Serial.begin(9600);
}

void loop() {

  ldrValue = analogRead(ldrPin);   // Read LDR value
  pirState = digitalRead(pirPin);  // Read PIR sensor

  Serial.print("LDR Value: ");
  Serial.print(ldrValue);
  Serial.print("  PIR: ");
  Serial.println(pirState);

  // Check for night condition
  if (ldrValue < 500) {   // Night time
    if (pirState == HIGH) {
      // Motion detected → High brightness
      analogWrite(ledPin, 255);
    }
    else {
      // No motion → Low brightness
      analogWrite(ledPin, 80);
    }
  }
  else {
    // Day time → LED OFF
    analogWrite(ledPin, 0);
  }
  delay(500);   // Small delay for stability
}
```



Scenario-2 : Smart Parking System

- **Scenario :** A smart parking system is required to monitor multiple parking slots automatically. Each parking slot should be able to detect the presence of a vehicle and indicate whether the slot is occupied or free. The system should also maintain and display the total number of available parking slots in real time.

- **Requirements**
  (a) Ultrasonic sensors to detect vehicle presence in each parking slot
  (b) LEDs to indicate slot status (occupied / free)

- **Programming Task :** Write program that performs the following tasks:
  (a) Measures the distance using ultrasonic sensors for each parking slot
  (b) Detects vehicle presence based on a predefined distance threshold
  (c) Turns ON the corresponding LED when a parking slot is occupied
  (d) Turns OFF the LED when a parking slot is free
  (e) Counts the total number of available parking slots
  (f) Displays the status of each slot and the total number of free slots on the Serial Monitor

Scenario-2 : Smart Parking System - Solution

```
#define slots 2

int trigPins[slots] = {3, 5};
int echoPins[slots] = {2, 4};
int ledPins[slots]  = {9, 12};

long duration;
int distance;

void setup() {
  Serial.begin(9600);

  for (int i = 0; i < slots; i++) {
    pinMode(trigPins[i], OUTPUT);
    pinMode(echoPins[i], INPUT);
    pinMode(ledPins[i], OUTPUT);
  }
}

void loop() {
  int freeSlots = 0;

  Serial.println("---- Parking Status ----");

  for (int i = 0; i < slots; i++) {
    digitalWrite(trigPins[i], LOW);
    delayMicroseconds(2);
    digitalWrite(trigPins[i], HIGH);
    delayMicroseconds(10);
    digitalWrite(trigPins[i], LOW);

    duration = pulseIn(echoPins[i], HIGH);
    distance = duration * 0.034 / 2;

    Serial.print("Slot ");
    Serial.print(i + 1);
    Serial.print(": Distance = ");
    Serial.print(distance);
    Serial.print(" cm → ");
```




```
    if (distance < 15) {
      digitalWrite(ledPins[i], HIGH);
      Serial.println("OCCUPIED");
    } else {
      digitalWrite(ledPins[i], LOW);
      Serial.println("FREE");
      freeSlots++;
    }
  }

  Serial.print("Total Free Slots: ");
  Serial.println(freeSlots);
  Serial.println("-----------------------");

  delay(3000);
}
```

Scenario-3 : Multi-Zone Temperature Monitoring

- **Scenario :** A smart building has three rooms, each monitored by a DHT11 temperature sensor. To avoid false alarms due to temporary temperature spikes in a single room, the system should raise an alert only when overheating is detected across multiple rooms for a sustained period.

- **Requirements**
  (a) Use three DHT22 sensors, each connected to a different digital pin
  (b) Read temperature values every 2 seconds
  (c) Temperature threshold: 32 °C
  (d) Trigger alert only if two or more rooms exceed the threshold for three consecutive readings
  (e) Display output on Serial Monitor

- **Programming Task:** Write an Arduino program that:
  (a) Reads temperature from all three DHT11 sensors
  (b) Counts rooms exceeding the threshold
  (c) Implements persistence logic using counters
  (d) Displays “BUILDING OVERHEAT” or “NORMAL” accordingly

Scenario-3 : Multi-Zone Temperature Monitoring

```ino
#include <DHT.h>

DHT dht[] = { DHT(2, DHT22), DHT(3, DHT22), DHT(4, DHT22) };

int hotCount = 0;

void setup() {
  Serial.begin(9600);
  for (int i = 0; i < 3; i++)
    dht[i].begin();
}

void loop() {
  int hotRooms = 0;

  for (int i = 0; i < 3; i++) {
    float t = dht[i].readTemperature();
    if (!isnan(t) && t > 32)
      hotRooms++;
  }

  if (hotRooms >= 2)
    hotCount++;
  else
    hotCount = 0;

  if (hotCount >= 3)
    Serial.println("BUILDING OVERHEAT");
  else
    Serial.println("NORMAL");

  delay(2000);
}
```